/**
 * Copyright (c) 2009 Cycle Information Technology
 *
 * Permission is hereby granted, free of charge, to any person obtaining a copy of this 
 * software and associated documentation files (the "Software"), to deal in the Software 
 * without restriction, including without limitation the rights to use, copy, modify, 
 * merge, publish, distribute, sublicense, and/or sell copies of the Software, and to 
 * permit persons to whom the Software is furnished to do so, subject to the following 
 * conditions:
 * 
 * The above copyright notice and this permission notice shall be included in all copies 
 * or substantial portions of the Software.
 * 
 * THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND, EXPRESS OR IMPLIED, 
 * INCLUDING BUT NOT LIMITED TO THE WARRANTIES OF MERCHANTABILITY, FITNESS FOR A 
 * PARTICULAR PURPOSE AND NONINFRINGEMENT. IN NO EVENT SHALL THE AUTHORS OR COPYRIGHT 
 * HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER LIABILITY, WHETHER IN AN ACTION OF 
 * CONTRACT, TORT OR OTHERWISE, ARISING FROM, OUT OF OR IN CONNECTION WITH THE SOFTWARE 
 * OR THE USE OR OTHER DEALINGS IN THE SOFTWARE.
 * 
 * Este comando es el motor del framework, su responsabilidad es detectar los cambios de estado de la aplicación
 * y realizar la notificación de generación de vista, al componente visual que corresponda.
 */
package com.cycleit.cycleframework.command {
	import com.cycleit.cycleframework.event.CycleEvent;
	import com.cycleit.cycleframework.event.ViewStateChangeEvent;
	import com.cycleit.cycleframework.interfaces.ICycleCommand;
	import com.cycleit.cycleframework.model.NavigationModel;
	
	/**
	 * En este comando se engloba toda la lógica de navegación de la aplicación en su nivel
	 * más básico posible, indicando qué comando ejecutar y cuándo, dependiendo del estado
	 * de navegación.
	 */
	public class ChangeViewCommand implements ICycleCommand{
		
		private var modelo:NavigationModel = NavigationModel.getInstance();
		/**
		 * Método <code>execute</code> que lanza la operación sin conocer el contenido de esta siguiendo el patrón de diseño: comando.
		 * 
		 * <p>1. Actualizamos el estado antiguo con lo que era nuestro estado actual.</p>
		 * <p>2. Generamos el nuevo <code>currentState</code>.</p>
		 * <p>3. Buscamos el componente visual desde el que se tiene que generar el nuevo estado y
		 *    lo notificamos.</p>
		 * 
		 * @param event El evento despachado que provocó la ejecución de este método.
		 * 
		 */		
		public function execute(event:CycleEvent):void {
			modelo.oldState = [].concat(modelo.currentState);
			//obtenemos el nuevo estado en funcion de los datos del evento
			var e:ViewStateChangeEvent = ViewStateChangeEvent(event);
			getNewState(e.state,e.index);
			
			notifyChange();
		}
		/**
		 * Generamos el nuevo estado actual, calculado en funcion del estado antiguo,
		 * la nueva cadena de identificadores, y el indice de navegacion desde el 
		 * que debemos de modificar los datos.
		 *  
		 * @param state El nuevo estado visual al cuál queremos navegar
		 * @param index Indica el índice a partir del cual debemos generar la nueva ruta de navegación dentro del estado actual.
		 * 
		 */		
		private function getNewState(state:Array, index:Number):void {
			if(index == 0){
				modelo.currentState = [].concat(state);
			}else{
				modelo.currentState = modelo.currentState.splice(0,index).concat(state);
			}
		}
		
		/**
		 * Aquí buscamos que componente visual de los que tenemos instanciados en la vista, es el encargado de<br>
		 * generar la vista de nuevo. El nivel anterior, al primer nivel que cambia, es el encargado<br>
		 * de regenerar las nuevas pantallas ya que el es el que tiene la información.<br>
		 * 
		 * Finalmente solo tenemos que notificar la generación activando StateControlInfo.
		 * 
		 */		
		private function notifyChange():void {
			if (modelo.currentState.length > 0) { 
				var i:int = 0;
				var cambiada:Boolean = false;
				
				var actualLength:uint = modelo.currentState.length;
				var oldLength:uint = modelo.oldState.length;
				while((i<actualLength)&&(i<oldLength)&&(!cambiada)){
					if(modelo.currentState[i] != modelo.oldState[i]){
						cambiada = true;
					}
					else {
						i++;
					}
				}
				if(!cambiada &&(actualLength != oldLength)){
					if(actualLength > oldLength){
						if(oldLength>0){
							modelo.navigationFlags[oldLength-1].changes = true;
						}else{
							modelo.navigationFlags[i].changes = true
						}
					}else{
						modelo.navigationFlags[actualLength-1].changes = true;
					}
				}else if(cambiada && i>0){
					//es menos dos por el incremento de la "i" al salir del bucle
					modelo.navigationFlags[i-1].changes = true;
				}
			}
		}
	}
}